/*
 *  Licensed to the Apache Software Foundation (ASF) under one
 *  or more contributor license agreements.  See the NOTICE file
 *  distributed with this work for additional information
 *  regarding copyright ownership.  The ASF licenses this file
 *  to you under the Apache License, Version 2.0 (the
 *  "License"); you may not use this file except in compliance
 *  with the License.  You may obtain a copy of the License at
 *
 *    https://www.apache.org/licenses/LICENSE-2.0
 *
 *  Unless required by applicable law or agreed to in writing,
 *  software distributed under the License is distributed on an
 *  "AS IS" BASIS, WITHOUT WARRANTIES OR CONDITIONS OF ANY
 *  KIND, either express or implied.  See the License for the
 *  specific language governing permissions and limitations
 *  under the License.
 */
pluginManagement {
    includeBuild('../grails-gradle') {
        name = 'grails-gradle'
    }
    includeBuild('../build-logic') {
        name = 'build-logic'
    }
    repositories {
        // because our publish plugin uses other plugins, those dependencies have to be resolvable
        // mavenLocal()
        mavenCentral()
        gradlePluginPortal()
        maven {
            url = 'https://repository.apache.org/content/groups/snapshots'
            content {
                includeVersionByRegex('org[.]apache[.]grails[.]gradle.*', '.*', '.*-SNAPSHOT')
            }
            mavenContent {
                snapshotsOnly()
            }
        }
        maven {
            url = 'https://central.sonatype.com/repository/maven-snapshots'
            content {
                includeVersionByRegex('cloud[.]wondrify.*', '.*', '.*-SNAPSHOT')
            }
            mavenContent {
                snapshotsOnly()
            }
        }
        maven {
            url = 'https://repository.apache.org/content/groups/staging'
            content {
                includeModuleByRegex('org[.]apache[.]grails[.]gradle', 'grails-publish')
            }
            mavenContent {
                releasesOnly()
            }
        }
    }
}
plugins {
    id 'com.gradle.develocity' version '4.1.1'
    id 'com.gradle.common-custom-user-data-gradle-plugin' version '2.3'
}

def isCI = System.getenv().containsKey('CI')
def isLocal = !isCI
def isReproducibleBuild = System.getenv('SOURCE_DATE_EPOCH') != null

develocity {
    server = 'https://ge.grails.org'
    buildScan {
        tag('grails')
        tag('grails-forge')
        publishing.onlyIf { it.authenticated }
        uploadInBackground = isLocal
    }
}

buildCache {
    local { enabled = (isLocal && !isReproducibleBuild) || (isCI && isReproducibleBuild) }
    remote(develocity.buildCache) {
        push = isCI
        enabled = !isReproducibleBuild
    }
}

rootProject.name = 'grails-forge'

include(
        'grails-cli',
        'grails-cli-shadow', // to allow us to transform our own artifacts
        'grails-forge-analytics-postgres',
        'grails-forge-api',
        'grails-forge-cli',
        'grails-forge-core',
        'grails-forge-web-netty',
        'test-core',
)

includeBuild('..') {
    name = 'grails-core'
}
includeBuild('../grails-gradle') {
    name = 'grails-gradle'
}

dependencyResolutionManagement {
    repositoriesMode.set(RepositoriesMode.FAIL_ON_PROJECT_REPOS)
    repositories {
        // mavenLocal() // Keep, this will be uncommented and used by CI (groovy-joint-workflow)
        mavenCentral()
        maven { url = 'https://repo.grails.org/grails/restricted' }
        maven {
            url = 'https://repository.apache.org/content/groups/snapshots'
            content {
                includeVersionByRegex('org[.]apache[.]grails.*', '.*', '.*-SNAPSHOT')
                includeVersionByRegex('org[.]apache[.]groovy.*', '.*', '.*-SNAPSHOT')
            }
        }
        maven {
            url = 'https://repository.apache.org/content/groups/staging'
            content {
                includeGroupByRegex('org[.]apache[.]grails.*')
            }
        }
        maven {
            url = 'https://central.sonatype.com/repository/maven-snapshots'
            content {
                includeVersionByRegex('cloud[.]wondrify.*', '.*', '.*-SNAPSHOT')
            }
            mavenContent {
                snapshotsOnly()
            }
        }
        maven {
            url = 'https://repository.apache.org/content/groups/staging'
            content {
                includeModuleByRegex('org[.]apache[.]grails[.]gradle', 'grails-publish')
            }
            mavenContent {
                releasesOnly()
            }
        }
    }
}
